
1. Bezpiecze ństwo - BHP

Przed  przystąpieniem do  zajęć  student  musi  bezwzględnie  zapoznać  się  z  instrukcja  opisującą
stanowisko  EMM.  Bez  odpowiedniej  wiedzy  i  znajomości  zagadnienia  nie  zostanie  dopuszczony  do
wykonywania ćwiczenia. 

Proces włączania i uruchamiania stanowiska przeprowadza jedynie prowadzący! Stanowisko gotowe
do zajęć ma uruchomiony komputer w którym pisze się i uruchamia programy. W przypadku niejasności, nie
naleŜy uruchamiać  programu na stanowisku tylko sprawdzić  informację  w dokumentacji. W razie braku

informacji w materiałach naleŜy spytać  się  prowadzącego. Klawisz  [ESC]
* na klawiaturze PC zatrzyma

wykonywanie programu.
Nie naleŜy przekraczać ustawionych osłon modułu gdy znajduje się on w ruchu. Nie moŜna wkładać

rąk i innych części ciała oraz innych przedmiotów (nie będącymi elementami wyposaŜenia) w zasięg pracy
robota  i  części  ruchomych  stanowiska.  W  razie  wystąpienia  niebezpieczeństwa  zdrowia,  kolizji  lub
nieprawidłowej  pracy stanowiska  naleŜy  nacisnąć  dowolny  awaryjny  wyłącznik  w postaci  czerwonego
przycisku “grzyba” , które są umieszczone w skrajnych częściach oraz pośrodku stanowiska. O zaistniałej
sytuacji naleŜy natychmiast powiadomić prowadzącego ćwiczenie.

NajbliŜsza pomoc medyczna to Przychodnia w budynku NT przy ulicy Narbutta 85, zaś  apteczka
pierwszej pomocy znajduje się na hali warsztatu

2. Programowanie

Programy sterujące pracą  modułu montaŜowego pisać  moŜna w dowolnym edytorze tekstu ASCII.
Zapis  programu  do  wykonania  na  stanowisku  musi  odbywać  się  w  postaci  pliku  tekstowego.  NaleŜy
stosować  nazwy  plików  tekstowych  składające  się  do  ośmiu  znaków,  bez  polskich  liter,  zakończone
rozszerzeniem *.prg

2.1 Pierwszy program – wiadomo ści podstawowe

Pierwszy program, nazwany jako witaj.prg to pierwsze kroki. W edytorze wpisano 3 poniŜsze linie:

# program wyswietlajacy napis Witaj
write "Witaj"
newl
stop

które tworzą pełny program. Objaśnienie, w kolejności linii (wierszy) programu:

1:

# - znak ten oznacza, Ŝe dana linia jest jedynie komentarzem, który nie ma wpływu na działanie
programu. Komentarz obejmuje całą jedną linię. Nie moŜna dodawać komentarzy do linii
zawierających polecenia. Jak widać powyŜej nie stosuje się polskich liter w kodzie źródłowym
programu.

2: instrukcja write wypisuje na ekran wartość wyraŜenia lub stałą tekstową, patrz punkt 3.3

3: przeniesienie wyświetlania do nowego wiersza

4:
instrukcja stop powoduje natychmiastowe zakończenie wykonywania programu i jej wpisanie w
ostatniej linii programu jest konieczne i niezbędne do prawidłowego działania programu !!! 
W kaŜdym programie powinna być przynajmniej jedna instrukcja stop.

Program składa się z linii tekstu zawierających dyrektywy i polecenia języka. W jednej linii moŜe znajdować
się tylko jedno polecenie, dyrektywa, komentarz lub etykieta. Linia moŜe mięć do 127 znaków długości.
Separatory (znaki spacji i tabulacji) mogą występować wewnątrz linii, na jej początku i na końcu. W

* Klawisze klawiatury komputerowej są oznaczane poprzez ujęcie ich nazwy w nawiasy kwadratowe.

przykładach nie stosowano innych znaków (np. podkreślenia) zamiast spacji. Poszczególne elementy (stałe,
identyfikatory, słowa kluczowe, operatory, etykiety) muszą być oddzielane separatorami. Nie istnieje
mechanizm pozwalający na kontynuacje jednej linii programu w następnej linii pliku - linia programu
stanowi całość w ramach linii pliku. Wielkość liter jest znacząca; tzn. identyfikatory 'Aaa' i 'aaa' są róŜne.

Dobrze: Źle:

write "Koniec" write"Koniec"

let ld01 4 let ld01_4

while ld02 :etykieta_3 while ld02:etykieta 3

gdzie:

1:
instrukcja ma na celu wypisanie na ekranie
słowa Koniec, instrukcja write i “koniec” są
rozdzielone spacją

brak rozdzielenie jest powodem błędu i nie
osiągnięcia celu

2:

instrukcja let przypisuje zmiennej ld01
wartość 4; w ten sposób wprowadza się w
programie nową zmienną, nadając jej
jednocześnie jakąś wartość,

instrukcja zamiast definiować zmienną i jej wartość
co jest w przypadku tego języka programowania
nierozerwalne wprowadza zmienną ld01_4 bez
wartości,

3:

dopóki wartość zmiennej ld02 jest róŜna od
zera wykonywane są polecenia zawarte
między tym wierszem a linią zawierającą
wyraŜenie :etykieta_3 (patrz 3.2.4),

między nazwą zmiennej ld02 i :etykieta brak spacji
powoduje, Ŝe interpreter rozumie ld02:etykieta (bez
3) jako nazwę zmiennej; oddzielona cyfra 3
powoduje, Ŝe cała linia jest niezrozumiała dla
interpretera (błąd programu)

Objaśnienia:
• polecenie – inaczej instrukcja,
• dyrektywa – zapis taki sam jako instrukcji, lecz na etapie kompilowania programu wykonywana jest

podczas wczytywania programu,
• komentarz –  wiersz lub wiersze  ułatwiające programiście  orientację  w działaniach programu i

celach poszczególnych jego części,
• etykieta – oznaczeniem miejsca w programie, np do którego ma nastąpić przekierowanie.

2.2 Uruchamianie programu

2.2.1 EMM

Praca  uŜytkownika  EMM  polega  na  obsłudze  komputera  PC  z  nakładką  ekranową  Norton
Commander(R) pracującego w systemie DOS - zalecana znajomość  instrukcji  obsługi. Proces włączania
stanowiska  EMM  przeprowadza jedynie  prowadzący! Stanowisko  gotowe do zajęć  ma  uruchomiony
komputer w którym pisze się i uruchamia programy. Zapis plików na dysku tylko w przeznaczonym do tego
miejscu C:\student\ lub w folderze wskazanym przez prowadzącego zajęcia.

W celu wykonania programu witaj.prg po akceptacji prowadzącego naleŜy:
1. najechać kursorem na plik w nakładce ekranowej Norton Commander(R) i nacisnąć [Enter] .
2. zatwierdzić  ponownym naciśnięciem klawisza [Enter]  domyślną  opcję  transpo lub wybrać  inną  z

podręcznego menu.  Opcja  transpo – kompilacja bez obsługi  sterowania robotem i  długotrwałej
procedury jego bazowania. Opcja interpretera z bazowaniem - robster.

2.2.2 Stanowiska studenckie

W trakcie realizacji ćwiczenia zespoły lub indywidualnie studenci pracują na stanowiskach PC1 – PC4
połączonych przez sieć komputerową z EMM. Dostęp do komputera modułu poprzez odpowiedni katalog na
zmapowanym dysku  G:  z np.  Komputer PC1  powinien pracować  w katalogu  1  itp. UŜytkowanie innych
katalogów niŜ  przyznane jest zabronione. W przypadku pracy na stanowisku studenckim, a takŜe w czasie

pracy na EMM bez nakładki, naleŜy uruchomić program wpisując nazwę interpretera oraz nazwę programu
jako parametr: 

transpo.exe witaj.prg [enter]

W przypadku gdy program nie znajduje się  w tym samym katalogu co interpreter naleŜy nazwę  programu
poprzedzić ścieŜka dostępu np.

transpo.exe c:\student\program\witaj.prg [enter]

W systemie Windows(R) ikonę pliku tekstowego *.prg moŜna przesunąć nad ikonę programu transpo
w oknie Eksploatora. MoŜliwe jest równie uruchomienie interpretatora poleceń  a następnie podanie nazwy
programu lub ścieŜki i nazwy programu. Konieczne jest tak jak i w poprzednim przypadku podanie pełnej
nazwy z rozszerzeniem programu w postaci *.prg

Realizację  programu  moŜna  w  dowolnej  chwili  przerwać  naciskając  klawisz  [ESC].  Wykonanie
programu rozpoczyna się od wykonania treści pierwszej linii programu. O ile polecenia nie wskazują inaczej
linie programu wykonywane są  kolejno według kolejności ich występowania w pliku / plikach programu.
Program zapisany jest w pliku / plikach ASCII. 

2.3 Podprogramy

W  przypadku  korzystania  z  podprogramu  uŜywa  się  jedynej  występującej  w  języku  dyrektywy
include. Składnia jej jest następująca:

include nazwa_pliku

gdzie nazwa_pliku jest poprawna nazwa pliku w systemie MS-DOS, np.

include plik.prg
include c:\prg\lib\uk_trans.prg

Dyrektywa powoduje dołączenie w miejscu jej wystąpienia do bieŜącego pliku programu zawartości
pliku podanego jako argument dyrektywy. Dyrektywy include mogą być zagnieŜdŜone i mogą występować
w plikach wielokrotnie. PoniŜej zamieszczono przykład programu wykorzystującego podprogram:

# program wykorzystujacy podprogram witaj.prg
include witaj.prg
newl
write "Koniec programu glownego"
wait 12
stop

gdzie:

1: komentarz

2:

wywołanie programu witaj.prg jako podprogramu, dyrektywa include powoduje, Ŝe podczas
uruchamiania programu kod źródłowy podprogramu jest “wstawiany” wewnątrz kodu źródłowego
programu, w miejscu, w którym jest wstawiona dyrektywa include; dyrektywy niezaleŜnie od
lokalizacji w kodzie źródłowym są wykonywane na etapie uruchamiania programu,

3: efektem instrukcji newl jest przejście do nowej linii wyświetlanych na ekranie danych,

4: instrukcja wypisania na ekranie 

5: czekaj określony czas, gdzie 1=0,25 sek.

6: zakończenie programu

Uwagi:
Po  uruchomieniu  powyŜszego  programu  nie  pokaŜe się  na  ekranie  napis:  Koniec  programu

głównego. Wynika to z faktu, Ŝe w  programie witaj.prg występuje instrukcja stop. W podprogramach
nie  stosuje  się  instrukcji  stop w  znaczeniu  końca  programu,  lecz  moŜe  ona  występować  np.  przy
instrukcjach  sterujących.  NaleŜy  je  jednak  uŜywać  gdy  wymaga  tego  algorytm  działania  programu.
Instrukcja  stop  występuje  najczęściej  w  programie  głównym  lub  samodzielnych  programach.  Po
poprawieniu programu witaj.prg, polegającym na usunięciu ostatniej linii zawierającej instrukcję  stop i
ponownym  uruchomieniu  całość  programu  zadziała.  Brak  instrukcji  stop w  programie  głównym  lub
samodzielnych programach spowoduje wyświetlenie błędu po wykonaniu wszystkich instrukcji w postaci
nazwy  pliku  z  ostatnim  ostatniej  linijki  programu  a takŜe  numerem  linii.  Instrukcja  newl  (skrót  od
NEWLine)  powoduje  przejście  do  nowej  linii  ekranu.  Brak  instrukcji  kasującej  cały  ekran  konsoli  i
ustawiającej  kursor  w  jego  lewym górnym rogu.  Instrukcja  wait  powoduje  zawieszenie  wykonywania
programu  (odczekanie)  liczby  ćwiartek  sekund  wskazanej  przez  wartość  wyraŜenia  podaną  po  spacji.
Składnia jest następująca:

wait wyraŜenie

Jeśli liczba ta jest ujemna, instrukcja  wait odczekuje 0 sekund. W naszym przypadku liczba 12/4 daje 3
sekundy oczekiwania.

3. Instrukcje

W języku występują cztery rodzaje instrukcji:
• instrukcje podstawienia wartości pod zmienną (występuje przy obliczeniach): let
• instrukcje sterujące: goto, goifplus, goifzero, call, return, stop, while, break,

continue,
• instrukcje wejścia / wyjścia: write, newl, read, wait, beep.

3.1 Instrukcje obliczeniowe

Instrukcja obliczeniowa let ma następująca składnię:

let zmienna wyrazenie

gdzie:
zmienna jest nazwą zmiennej (definiowanej i ewentualnie - przy pierwszym wystąpieniu nazwy zmiennej -
deklarowanej),  natomiast  wyrazenie  jest  dowolnym  wyraŜeniem  arytmetycznym.  Przykłady  podano
poniŜej.

3.1.1 Zmienne

W programie  moŜliwe  jest  uŜywanie  zmiennych.  Są  to  wyłącznie  globalne  zmienne proste  typu
rzeczywistego.  Identyfikatorem  (nazwą)  zmiennej  jest  dowolny  poprawny  identyfikator.  Zmienne  są
deklarowane (i jednocześnie definiowane) przez ich pierwsze wystąpienie jako 1 argumentu polecenia  let.
Zmienne  istnieją  aŜ  do  wykonania  polecenia  stop.  Odwołanie  się  do  nieistniejącej  (niezdefiniowanej)
zmiennej stanowi błąd. Przykład definiowania (deklarowania) zmiennej:

# deklaracja zmiennej a (jednoczesne nadanie wartosci 5)
let a 5

gdzie:

1: komentarz

2: deklaracja zmiennej o nazwie a, o zadanej wartości równej 5

Dopuszcza się przeciąŜenie identyfikatora jako jednocześnie nazwy zmiennej i nazwy etykiety, np.:

goifplus ident :ident

3.1.2 Operatory i wyra Ŝenia arytmetyczne

W  języku  zaimplementowano  poniŜsze  operatory  arytmetyczne,  stosując  zapis  wg  tzw.  Zasady
odwrotnej notacji polskiej (ONP).

Operator Liczba
argumentów Znaczenie: Przykład zapis

klasyczny: Przykład wg ONP:

* 2 mnoŜenie a*b a b *

/ 2 dzielenie a/b a b /

+ 2 dodawanie a+1 a 1 +

- 2 odejmowanie a-3 a 3 -

log 1 logarytm naturalny log 5 5 log

power 2 potęgowanie x^y 2^3 2 3 power

sin 1 sinus sin 0.7 0.7 sin

arctg 1 arc tangens arctg 0.7 0.7 arctg

Wszystkie operatory przyjmują argumenty rzeczywiste i zwracają wartości rzeczywiste. Wszystkie kąty
wyraŜane są w radianach. Brak jest operatorów unarnych (jednoargumentowych, np. -a) i operatorów
przetwarzających napisy.

WyraŜeniem arytmetycznym jest  sekwencja identyfikatorów zmiennych  liczbowych,  operatorów i
stałych  liczbowych  oddzielonych  separatorami.  W  wyraŜeniach  nie  stosuje  się  nawiasów  oraz  znaku
równości. 

Przykład 3.1

# deklaracja zmiennej wynik (jednoczesne nadanie wartosci 2+3)
let wynik 2 3 +
write "wynik = "
write wynik

gdzie:

1: komentarz

2: deklaracja zmiennej o nazwie wynik, której wartość wynosi 2 + 3

3: instrukcja wypisania na ekranie słowa i znaku “wynik = ”

4: instrukcja wypisania na ekranie wartości zmiennej wynik, a poniewaŜ brak jest instrukcji newl
wartość ta jest wypisana w tej samej linii co słowo i znak “wynik = ”

Przykład 3.2

# deklaracja zmiennej a (jednoczesne nadanie wartosci 2)
let a 2
write "potega "
write a
write " = "
write a 2 power

gdzie:

1: komentarz

2: deklaracja zmiennej o nazwie a, której wartość wynosi 2

3: instrukcja wypisania na ekranie słowa i spacji “potega ”

4: instrukcja wypisania na ekranie wartości zmiennej a, a poniewaŜ brak jest instrukcji newl wartość ta
jest wyświetlona na ekranie w tym samym wierszu co słowo “potega ”

5: instrukcja wypisania na ekranie spacji i znaków “ = “ i podobnie jak powyŜej brak instrukcji newl
powoduje wyświetlanie w tym samym wierszu

6:
obliczenie potęgi o wykładniku równym 2 zmiennej a i wypisanie wyniku obliczenia na ekranie w
tym samym wierszu, co dotychczas

3.2 Instrukcje steruj ące

goto 
goifplus 
goifzero
call 
return 
stop
while 
break 
continue

3.2.1 Instrukcje skoków

Instrukcja  goto powoduje  zmianę  kolejności  wykonywania  poleceń  programu  -  skok  do  linii
oznaczonej podaną etykietą. Składnia instrukcji jest następująca:

goto :etykieta

przy czym etykieta :etykieta musi istnieć w programie np.:

:poczatek
write "Witaj"
goto :poczatek

Uwaga:
PowyŜszy fragment kodu źródłowego program będzie wyświetlał  napis w sposób ciągle – jeden za

drugim aŜ  do końca linijki,  a następnie nastąpi przejście do nowego wiersza itd. Aby zakończyć  naleŜy
nacisnąć klawisz [Esc].

Instrukcja  goifplus powoduje  skok  do  linii  zawierającej  podana etykietę,  jeśli  wartość  podanej
zmiennej jest dodatnia. Składnia instrukcji goifplus jest następująca:

goifplus zmienna :etykieta
przy czym etykieta :etykieta musi istnieć w programie np.:

let a 2
goifplus a :dodatnie
write "a niedodatnie – bez skoku"
stop
:dodatnie
write "a dodatnie – dobrze"
stop

Uwaga:
Tylko  przy  niedodatnich  wartościach  deklarowanych  zmiennej  a pojawi  się  informacja  o  nie

wykonaniu skoku do etykiety i program się zakończy.

Instrukcja  goifzero powoduje skok  do  linii  zawierającej  podana  etykietę,  jeśli  wartość  podanej
zmiennej jest równa 0.0 Składnia instrukcji goifzero jest następująca:

goifzero zmienna :etykieta

przy czym etykieta :etykieta musi istnieć w programie np.:

# deklaracja zmiennej ld14 (jednoczesne nadanie wartosci 3, innej niŜ 0 i 1)
let ld14 3
# odczyt do zmiennej ld14 stanu portu wejsciowego nr 2 (wartosc 0 lub 1)
get ld14 2
goifzero ld14 :brak_chwytaka
goto :dalej
:brak_chwytaka
write "brak chwytaka na wieszaku"
stop
:dalej
write "chwytak na wieszaku"

Uwaga:
W  przypadku  braku  chwytaka  stan  portu  nr  2  (ld14)  wynosi  0  i  następuje  skok  do  etykiety

:brak_chwytaka ,
wyświetlenie komunikatu i zakończenie programu.

3.2.2 Etykiety

Etykiety są to identyfikatory poprzedzone znakiem “:” (dwukropek). SłuŜą do oznaczania okreśłonych
miejsc w kodzie źródłowym programu. Etykieta nie moŜe zawierać  separatora. Etykieta moŜe wystąpić  w
programie w dwóch postaciach:

• definiującej: jako określająca miejsce w programie np. :
:tu_skoczyc

• wskazującej: jako wskazanie na określone miejsce w programie np. :
goto :tu_skoczyc

Postać definiująca etykiety moŜe wystąpić tylko jeden raz w całym programie. Patrz teŜ informacje i
przykłady w instrukcjach skoków (3.2.1).

3.2.3 Instrukcje wywołania i powrotu z procedury

Instrukcja  call słuŜy  do  skoku  do  wskazanej  etykieta  linii  z  zapamiętaniem  miejsca  powrotu.
Instrukcja  return powoduje powrót  do  zapamiętanego  miejsca.  Obie  instrukcje  implementują  skok  ze
śladem pozwalający na tworzenie procedur (podprogramów). Procedury mogą być zagnieŜdŜone. W jednej
procedurze  moŜe  być  wiele  instrukcji  return.  Procedura  wewnątrz  kodu  źródłowego  programu  lub
podprogram  w  postaci  wydzielonego  i  zapisanego  w  oddzielnym  pliku  kodu  źródłowego  słuŜą  do
wielokrotnego  wykorzystywania.  Takie  rozwiązania  pozwalają  na  uporządkowanie  określonych  działań
programów i nie powtarzanie wielokrotne tego samego ciągu instrukcji w kodzie źródłowym. JeŜeli program
daną czynność ma wykonywać wielokrotnie to naleŜy dąŜyć do opracowania procedury lub podprogramu.
Ma to takŜe znaczenie w przypadku wprowadzania zmian, które nanosi się tylko w jednym miejscu, a nie w
wielu. Składnia powyŜszych instrukcji jest następująca:

call :etykieta
...

return

Z  wywołaniem  procedury  (tzn.  uŜyciem  polecenia  call)  nie  wiąŜą  się  Ŝadne  mechanizmy
przekazywania parametrów i tworzenia zmiennych lokalnych. Parametry moŜna przekazywać  przez tzw.
zmienną globalną. Instrukcja  stop powoduje natychmiastowe zakończenie wykonywania programu. Patrz
punkt 2.1 i 2.3

3.2.4 Instrukcja p ętli

Instrukcje  while,  break i  continue pozwalają  na organizacje pętli ze sprawdzaniem warunku na
początku. Składnia tych instrukcji jest następująca:

while zmienna :etykieta
...
#zakres pętli – miedzy instrukcja while a :etykieta
...
:etykieta
break
continue

Wykonanie instrukcji while polega na:

1:

sprawdzeniu czy wartość zmiennej sterującej zmienna jest róŜna od 0.0,
• jeśli tak, wykonywany jest zakres pętli aŜ do napotkania etykiety :etykieta, a następnie

wraca się ponownie do wiersza 1 kodu programu,
• jeśli nie, wykonanie pętli jest kończone i sterowanie przekazywane jest do linii następnej po

linii zawierającej etykietę :etykieta

2: początek zakresu pętli

3: komentarz 

4: koniec zakresu pętli

5: etykieta :etykieta

6: instrukcja break powoduje natychmiastowe przerwanie wykonywania zakresu pętli,

7: instrukcja continue polega na natychmiastowym porzuceniu wykonywania zakresu pętli i przejściu
do kolejnego sprawdzenia.

Pętle mogą  być  wzajemnie zagnieŜdŜone. Napotkane polecenia  break  i  continue  rozumiane są  jako
odnoszące się do aktualnie wykonywanej (najmniej zagnieŜdŜonej) pętli.

Przykład 3.3

# podprogram ruchu palety w lewo do stacji lewej
write "Ruch w lewo"
set 13
write "Zderzak lewej stacji aktywny"
set 4
# definicja zmiennej petli (najczesciej wartosc 1)
let stacja 1
# definicja zmiennej ld03 (nadanie wartosci 2, rozna od odczytanej 1 lub 0)
let ld03 2
# definicja petli (zmienna, etykieta)
while stacja :na_stacji
# odczyt portu 23 do zmiennej ld03
get ld03 23
# obliczenie zmiennej petli (stacja = stacja - ld03)
let stacja stacja ld03 -
# definicja etykiety wyjscia z petli
:na_stacji
clear 13
write " obroty stop"
newl
write "Paleta na stacji roboczej w gore"
set 0
wait 1
newl
write "Paleta podniesiona"

Przykład 3.4.a i b

z uŜyciem break z uŜyciem continue

# demo34a - polecenie break
let n 5
write "n="
write n
newl
while n :koniec
let n n 1 -
break
write "n="
write n
newl
:koniec
write "n="
write n
newl
# koniec

# demo34b - polecenie continue
let n 5
write "n="
write n
newl
while n :koniec
let n n 1 -
continue
write "n="
write n
newl
:koniec
write "n="
write n
newl
# koniec

programik wypisuje:
  n=5
  n=4

programik wypisuje:
  n=5
  n=0

3.3 Instrukcje wej ścia / wyj ścia konsolowego

Instrukcjami wejścia (klawiatura) / wyjścia (ekran, głośnik systemowy) konsolowego są  instrukcje o
następującej składni:

write wyrazenie
write "stala tekstowa"
newl
wait wyrazenie
beep wyrazenie
read zmienna
file

1:

2:
instrukcja write wysyła na konsole wartość wyraŜenia lub stałą tekstową (tekst),

3: instrukcja newl wysyła na konsolę znak przejścia do nowej linii,

4: instrukcja wait odczekuje podaną jako wyrazenie liczbę ćwiartek sekundy,

5: instrukcja beep generuje dźwięk trwający podaną jako wyrazenie liczbę ćwiartek sekundy,

6:
instrukcja read wczytuje z klawiatury liczbę i jej wartość przypisuje zmiennej zmienna, która
musi być wcześniej zdeklarowana; gdy odczyt nie odbył się poprawnie, następuje błąd i przerwanie
pracy programu

7: instrukcja file 

Przykład 3.5
#program potega
:start
# definicja etykiety
newl
# przejscie do nowej linii
write "Podaj liczbe podnoszona do potegi: "
let liczba 0
# definicja zmiennej z nadaniem wartosci 0
read liczba
#odczyt zmiennej liczba
let potega 0
# definicja zmiennej z nadaniem wartosci 0
let potega liczba 2 power

# obliczenie potegi (potega = liczba ^2)
write "Potega liczby "
write liczba
write " wynosi "
write potega
beep 1
wait 2
beep 1
# sygnal dzwiekowy podwojny
goto :start
# skok do etykiety

3.3.1 Procedura pauza

Procedura  słuŜy  do  okresowego  zatrzymywania  wykonania  programu  w czasie   jego  testowania.
Wywołanie procedury skutkuje komunikatem na ekranie i oczekiwaniem na naciśniecie klawisza [Enter].

Składnia tej procedury jest następująca:

call :pauza
lub

let pauza parametr
call :pauza

parametr

Parametrem (opcjonalnym) jest numer pauzy. W wypadku wielu wywołań
procedury parametr pozwala zorientować się o które wywołanie chodzi.
Parametr przekazuje sie w zmiennej pauza. 
Wartość parametru pozostaje niezmienna az do kolejnej - jawnej – zmiany  w
programie

3.4 Instrukcje freemem

Instrukcja  freemem powoduje  wykasowanie  wszystkich  zmiennych  istniejących  w  programie.
Wartości zmiennych są tracone, a ich identyfikatory stają się nieznane (wszystkie zmienne przestają istnieć -
sytuacja jest taka, jak przy uruchomieniu programu).

Składnia instrukcji freemem jest nastepująca:

freemem

Instrukcja przeznaczona jest do współpracy z instrukcją spawn.

3.5 Instrukcje spawn

Instrukcja  spawn powoduje  przerwanie  wykonywania  bieŜącego  programu  i  załadowanie  programu
podanego jako agrument. Istniejące w programie zmienne zostają zachowane. Nazwę pliku podaje się w
cudzysłowiu. Przy wykonaniu instrukcji spawn stan systemu (robota, układu transportowego, portów, itd)
pozostaje niezmieniony.

Skladnia instrukcji spawn jest nastepująca:

spawn ”nazwa_pliku”
  
np. 

spawn "c:\pliki\program2.prg"

4. Stanowisko

Do instrukcji sterujących stanowiska zalicza się instrukcje set i clear. Składnia ich jest następująca:

set wyrazenie
clear wyrazenie

gdzie:
wyraŜenie wyrazenie wyznacza numer portu wyjściowego (od 0 do 14) patrz (4.4).
Instrukcje set i clear odpowiednio ustawiają stan wskazanego przez wyrazenie portu wyjściowego na ”1”
lub ”0”.

Przykład 4.1

# podprogram wystawia zderzak stacji kodowania i chowa go po 2 sekundach
set 2
# ustawienie "1" na porcie wyjsciowym 2
wait 8
# czas oczekiwania
clear 2
# wyzerowanie portu wyjściowego 2

Uwaga:
Przy  programowaniu  naleŜy  uwzględnić  czas  potrzebny  na  przetworzenie  sygnału  sterującego  w

urządzeniu  wykonawczym.  PrzewaŜnie  okres  zadziałania  wynosi  0,25  –  1  sekundy  w  zaleŜności  od
wielkości elementu wykonującego ruch.

4.1 System transportowy

Rozmieszczenie elementów systemu transportowego określone są  w części  1 instrukcji  opisującej
budowę stanowiska.

4.1.1 Zderzaki

Zderzaki są jednokierunkowe – zatrzymują paletę na stacji tylko przy odpowiednim kierunku najazdu
palety. Przy niewłaściwym kierunku ruchu paleta nie zatrzyma się na stacji tylko w jej pobliŜu. Zderzaki po
wysterowaniu stanowią przeszkodę w ruchu palet w obu kierunkach.

set 2 zderzak stacji kodującej w górze,

set 3 zderzak stacji prawej w górze,

set 4 zderzak stacji lewej w górze,

set 5 zderzak stacji końcowej (odkładczej, po lewej stronie) w górze,

set 12 zderzak stacji odczytu w dole (umoŜliwiony przejazd palety)

Zderzaki po wysterowaniu są aktywne (zatrzymują palety - wyjątek zderzak stacji odczytu). Zderzaki
zatrzymują  palety na stacjach ze swojej prawej strony, tj. unieruchomienie palety na odpowiedniej pozycji
odbywa się  przy ruchu jej w lewo) - wyjątek stanowi zderzak końcowy stanowiska odkładczego z lewej
strony stanowiska.

4.1.2 Podniesienie stacji

set 0 stacja robocza lewa i prawa jednocześnie uniesione (pozycjonowanie)

set 13
set 14

uniesienie stacji załadowczo-wyładowczej (po prawej stronie stanowiska), naleŜy
stosować obie komendy jednocześnie, przy czym wpisanie obu komend musi
następować w kolejnych liniach

Przykład 4.2

# podprogram unoszacy palete na stacji zaladowczo-wyladowczej
set 13
set 14
write "Paleta na stacji zal-wyl uniesiona"

Do podniesienia stacji zał-wył  wykorzystujemy właściwość  systemu transportowego polegającą  na
braku jednoczesnego występowania “1” na obu wspomnianych wyjściach portu.  Wynika to z uwagi  na
oczywistą  niemoŜność  równoczesnego  ruchu  w  dwóch  kierunkach.  Kombinacja tych  dwóch  sygnałów
sterujących  “1”  jest  wykorzystywana  do  wysterowania  podnośnika  stacji  zał-wył  bez  konieczności
zwiększania  ilości  sygnałów  (np.  poprzez  multiplikację)  przy  ograniczonych  zasobach  systemowych
sterownika PLC. 

4.1.3 Kodowanie

Do  identyfikacji  palet  wykorzystywany  jest  kod  ustawiany  w  kostce  kodowej  zamontowanej  na
palecie. Odpowiedni kod moŜe być przypisany np. odpowiedniemu rozmieszczeniu elementów na palecie.

set 8 kodowanie palety – trzpień 3 wysunięty,

set 9 kodowanie palety – trzpień 4 wysunięty,

set 10 kodowanie palety – trzpień 1 wysunięty,

set 11 kodowanie palety – trzpień 2 wysunięty,

clear 8 kodowanie palety – trzpień 3 cofnięty,

clear 9 kodowanie palety – trzpień 4 cofnięty,

clear 10 kodowanie palety – trzpień 1 cofnięty,

clear 11 kodowanie palety – trzpień 2 cofnięty,

Kod palety moŜe zostać  nadany jedynie zatrzymanej palecie na stacji kodowania. Proces kodowania
polega na pojedynczym wysunięciu trzpienia z pary i  zwłoce czasowej przed komendą  chowającą  dany
trzpień.  Potem naleŜy wysunąć  kolejny trzpień  z pary nadających określony kod. Kolejność  wysuwania
trzpieni dowolna.

kod palety 10 trzpień 1 i 3

kod palety 9 trzpień 2 i 3

kod palety 6 trzpień 1 i 4

kod palety 5 trzpień 2 i 4

Przykład 4.3

# podprogram ustawia kod 5
set 9
wait 1
clear 9
set 11
wait 1
# ciag dalszy listingu na nastepnej stronie

clear 11
wait 1
write "ustawiony kod palety 5 "

Kodowanie palety odbywa się  na stacji kodowania w czasie spoczynku palet. Zwłoka czasowa po
komendzie  wysuwającej  trzpień  umoŜliwia  jego  ruch.  Brak  opóźnienia  nie  pozwala  przemieścić  się
trzpieniowi do pozycji kodującej. Identyczna sytuacja jest przy jego cofaniu. Przed przemieszczeniem palety
ze  stacji  kodowania  naleŜy  upewnić  się  czy  w programie  znajdują  się  komendy chowające  trzpienie i
występuje zwłoka czasowa na ruch powrotny trzpieni. W przeciwnym wypadku trzpienie działać  będą  jak
zderzak. Z uwagi na mniejsze ich wymiary niŜ zderzak spowoduje to uszkodzenie głowicy kodującej.

Kodowanie palety moŜna przeprowadzić takŜe z zastosowaniem procedury setkodpal, która posiada
następującą składnię:

let kodpal parametr
call :setkodpal

parametr parametrem wywołania jest poprawny kod (tj. ustawialny na palecie), tzn.: 5, 6, 9,
10

Uwagi:
• kody inne niŜ poprawne są ignorowane,
• procedura nie działa, gdy na stacji kodowania nie ma palety.

4.1.4 Odczyt kodu

Na podstawie sygnałów z stacji odczytu odpowiadających wysuniętym poszczególnym trzpieniom w
kostce kodowej moŜna zidentyfikować  paletę. Paleta w czasie odczytu kodu palety musi znajdować  się  na
stacji  odczytu.  Kod  palety  w  postaci  dziesiętnej  otrzymywany  jest  przypisując  odczytanym  sygnałom
następujące wagi:

kod_palety = trzpień_4 * 8 + trzpień_3 * 4 + trzpień_2 * 2 + trzpień_1 * 1
co odpowiada
kod_palety = ( ( trzpień_4 * 2 + trzpień_3 ) * 2 + trzpień_2 ) * 2 + trzpień_1 * 1

Przykład 4.4

# podprogram odczytuje aktualny kod palety
let kodpal 0
# laczniki drogowe LD poprzez podprogram ld_czyt.prg
include ld_czyt.prg
let kodpal ld10 2 * ld09 + 2 * ld08 + 2 * ld07 +
write "Kod palety = "
write kodpal

Odczyt kodu palety moŜna zrealizować poprzez procedurę getkodpal, której składnia jest następująca:

call :getkodpal

Procedura nie wykorzystuje Ŝadnych parametrów. W odpowiedzi na wywołanie procedura ta zwraca
wartość, która jest przekazywana w zmiennej globalnej kodpal.

kodpal MoŜliwe wartości to: 5, 6, 9 lub 10

Wartość zmiennej kodpal moŜe być dowolnie wykorzystywana, nie zaleca się jednak jej zmieniania.

Uwagi:
• procedura nie działa, gdy na stacji odczytu nie ma palety.

4.1.5 Przemieszczanie palet

set 13 ruch palety w prawo,

set 14 ruch palety w lewo,

Uwaga: Aby zatrzymać trwający ruch palet w prawo konieczna jest komenda

clear 13

Zmiana  kierunku  ruchu  palet  wymaga  miedzy  zatrzymaniem  ruchu  w  danym  kierunku  a
uruchomieniem przeciwnego ruchu 1 sekundowej przerwy czasowej na zatrzymanie się silnika.

Przykład 4.5

# program zmiana kierunku ruchu palet
set 13
# ruch palet w prawo
wait 10
# czas oczekiwania (ewentualnie odczyt z czujnika)
clear 13
# zatrzymanie ruchu w prawo palet
wait 4
# opoznienie wymagane przy zmianie kierunku!!!
set 14
# ruch palet w lewo
wait 10
# czas oczekiwania
clear 14
# zatrzymanie ruchu w prawo palet

Uwaga:
Gdy po komendzie  nakazującej  ruch w danym kierunku nie  odwołamy ruchu palet  a dodatkowo

polecimy  ruszać  się  transporterowi  w  przeciwnym  kierunku,  nastąpi  awaryjne  zatrzymanie  ruchu  i
podniesienie się stacji załadowczo-wyładowczej niezaleŜnie czy na stacji znajduje się paleta czy nie. NaleŜy
unikać tej sytuacji.

Ruch  palet  moŜe  być  zrealizowany  poprzez  uŜycie  procedury  ruchpal,  której  składania  jest
następująca:

let pal2poz parametr
call :ruchpal

gdzie jako parametr stosuje się:

1 stacja załadowczo-wyładowcza,

2 stacja kodowania,

3 stacja odczytu,

4 stacja robocza prawa,

5 stacja robocza lewa,

6 lewa skrajna stacja końcowa,

Uwagi:
• przed uruchomieniem programu wykorzystującego bibliotekę paleta  musi znajdować się na pozycji

1 (stacja załadowczo-wyładowcza),
• wywołanie procedury z innym parametrem niŜ 1, 2, 3, 4, 5 lub 6 powoduje zatrzymanie programu,
• wywołanie jałowe (np. z parametrem 3 gdy paleta stoi na stacji 3) są ignorowane.

Przykład 4.6

# przemieszczenie palety z pozycji dowolnej na pozycje 5 realizuje sekwencja
let pal2poz 5
call :ruchpal

4.2 Manipulator 

Przestrzeń  roboczą  określa  rysunek  1.  Podano  na  nim  równieŜ  współrzędne  chwytaków  na
wieszakach, pozycji bezpiecznej oraz pozycję kołków bazujących stacji roboczych.

Uwaga: Przestrzeń robocza układu symulatora i systemu EMM róŜnią się nieznacznie.

4.2.1 Ruch

Przemieszczenie kiści robota z chwytakiem lub bez niego moŜe dokonywać się jedynie w jednej osi na
raz. Do czasu zakończenia ruchu zatrzymane jest przetwarzanie kolejnych linii programu.

Instrukcja sterująca ruchem robota ma postać:

robotgo os pozycja

gdzie:

os
zmienna lub stała oznaczająca oś (1-3), 1-ruch w poprzek (lewo-prawo), 2-ruch w
głąb (przód-tył), 3-ruch wzdłuŜ (góra-dół) - patrz punkt 4.2.4,

pozycja wyraŜenie określające pozycje (połoŜenie) w danej osi w mm,

Przykład 4.7

# ruch w trzeciej osi (gora-dol) na pozycje 80
robotgo 3 80

Zakresy ruchów w poszczególnych osiach wynoszą.(rys.1):
Oś 1(x) - wartości od 0 do 690
Oś 2(y) - wartości 0-380
Oś 3(z) - wartości 0-80

Rys.1 

4.2.2 Manipulowane cz ęści

Robot moŜe przemieszczać dwa rodzaje części przedstawione na rysunku 2. 
Część  typu korek (1 sztuka) przenoszona jest za pomocą  chwytaka zewnętrznego (wyposaŜonego w

zewnętrzne nakładki szczęk). Korek moŜe być umieszczony bezpośrednio na palecie lub za pośrednictwem
tulejek. Do dyspozycji jest siedem części typu tuleja o wymiarach “L” równych: 12 (2 sztuki), 14 (1 sztuka),
16 (2 sztuki) i 20 (2 sztuki). Umieszczane są  one w gniazdach wykonanych na paletach. Patrz punkt 4.2.3
Budowa tulejek umoŜliwia umieszczenie jednej na drugiej. Uzyskana dzięki temu kombinacja wysokości
rozszerza moŜliwości  manipulacji częściami.  Do przemieszczania pojedynczych tulejek nie obsadzonych
korkiem wykorzystywany jest chwytak wewnętrzny.

Rys. 2

4.2.3 Chwytaki

set 7 chwytak zamocowany w kiści (przez zmieniacz chwytaków) robota,

set 6 otwórz szczęki chwytaka zamocowanego w kiści robota,

W celu zamknięcia szczęk chwytaka uchwyconego w kiści  (rysunek 3) naleŜy odwołać  komendę
otwierającą  poprzez  instrukcję  clear.  Odmocowanie  chwytaka  moŜe  odbyć  się  jedynie  po  dojeździe
bezpieczną trajektorią przez robota do wolnego wieszaka. Po odmocowaniu konieczny jest ruch w osi 3 na
współrzędną co najmniej 20 przed ruchem w pozostałych osiach lub zakończeniem programu.

Rys. 3

Pozycje chwytaków na wieszakach. Liczby określają  kolejne współrzędne bezwzględne w poszczególnych
osiach (oś 1, oś 2 i oś 3) w układzie EMM:

Wieszak lewy: Wieszak środkowy: Wieszak prawy:

480, 0, 0 581, 0, 0 681, 0, 0

Pobranie chwytaka pustą(!!!) ki ścią manipulatora następuje od góry - trzecia oś o wartości minimum
20,  następnie  wymagane  jest  osiąganie  dwóch  pozostałych  współrzędnych  i  dopiero  po  ich  uzyskaniu
obniŜenie pustej kiści w dół, do podanych wyŜej wartości współrzędnych odpowiednich wieszaków. Po
uchwyceniu chwytaka w kiści niezbędny jest ruch do przodu z uchwytu wieszaka w osi 2 na pozycję  15,
zapewniającej wysunięcie się  ze spręŜyn mocujących chwytak na wieszaku i umoŜliwiając ruch jedynie w
pionie. Ruch w poprzek (oś 1) moŜliwa jest dopiero po minięciu (przyrost wartości współrzędnych w osi 2 z
wartości 15 do 80) ramion wieszaka – niezaleŜnie od współrzędnej w osi 3.

Procedura  chwytak pozwala,  przy  zachowaniu  określonych  warunków  na  pominięcie  ustalania
dokładnego  przebiegu  ścieŜki  manipulatora  w  celu  pobrania  z  magazynu  Ŝądanego  chwytaka.
Automatycznie następuje przejazd chwytaka z dowolnej pozycji na pozycję określoną jako bezpieczną. W
kolejnych krokach następuje pobranie chwytaka  i  jego wyprowadzenie z  magazynu.  Procedura  posiada
następującą składnię:

let chwytak parametr
call :chwytak

gdzie jako parametr stosuje się:

0

odłóŜ chwytak,
jeŜeli w kiści manipulatora nie znajduje się Ŝaden chwytak to nic się nie stanie,
jeŜeli juŜ jakiś chwytak jest załoŜony, to zostanie odłoŜony na swój wieszak w
magazynie,

1

pobierz chwytak nr 1 (tzn. ten w wieszaku nr 1 - skrajnym lewym)
jeŜeli w kiści manipulatora jest juŜ chwytak nr 1, to nic się nie dzieje;
jeŜeli jest juŜ jakiś inny chwytak, to zostanie on odłoŜony na swoje miejsce i
pobrany zostanie chwytak nr 1

2 analogicznie jak dla parametru 1 (tylko wieszak środkowy)

3 analogicznie jak dla parametru 1 (tylko wieszak prawy)

Uwaga: 
• przed uruchomieniem programu wykorzystującego bibliotekę wszystkie  trzy chwytaki musza być w

wieszakach,
• uŜycie innego parametru niŜ 0, 1, 2 lub 3 zatrzymuje program,
• po pobraniu chwytaka robot staje na pozycji:

x=450/550/650 (zaleŜnie od chwytaka)
y=55
z=80

• procedurę naleŜy wywoływać w bezpiecznym połoŜeniu robota nad paleta, zalecane są następujące
współrzędne:
x=450/550/650 (zaleŜnie od chwytaka)
y=55
z=80

• za doprowadzenie w sposób bezpieczny do tego połoŜenia  odpowiada autor programu
wykorzystującego bibliotekę,

• błąd uchwycenia chwytaka przez kiść robota powoduje przerwanie programu.

Przykład 4.8.a i b
przykład z uŜyciem tzw. języka niŜszego rzędu przykład z uŜyciem tzw. języka wyŜszego rzędu

# demo 48a 
# pobranie chwytaka nr 3
# kisc pusta; robot na pozycji 500,30,45
robotgo 3 80
robotgo 2 0
robotgo 1 650
robotgo 3 0
set 7
wait 2
get ch_pom 10
goifplus ch_pom :jest_ok
write "Blad uchwycenia chwytaka"
newl
stop
:jest_ok
robotgo 2 20
robotgo 3 80
robotgo 2 55
stop
# robot na pozycji 650,55,80
# koniec

# demo 48b
# pobranie chwytaka nr 3
# kisc pusta; robot na pozycji 500,30,45
include robolib.prg
let chwytak 3
call :chwytak
stop
# robot na pozycji 650,55,80
# koniec

4.2.4 Współrz ędne na paletach

Rozmieszczenie gniazd w paletach zabazowanych na stacjach roboczych względem kołków bazowych
przedstawia rysunek 5 (str.22).

4.2.5 Współrz ędne manipulatora
Współrzędne osi 1 i 2 robota zgodne z punktem 4.2 i 4.2.2 Do uchwycenia manipulowanych części

naleŜy  zająć  robotem z  zamocowanym chwytakiem zewnętrznym lub  wewnętrznym  (wyposaŜonym w
odpowiednie nakładki szczęk) określona pozycję  nad paletą  w osi 3,  patrz rysunek 4. Punkt “0” robota
wysterowanego w osi 3 na współrzędna 80 znajduję się 207,5 mm nad powierzchnią zabazowanej palety.

Rys.4

Rys. 5 Schemat poglądowy przedstawiający punkt programowy, chwytak, paletę
a.) dane wg instrukcji i układu rzeczywistego, 
b.) szkic do wyznaczenia współrzędnej zq dla zadanej odległości q chwytaka od powierzchni
     zabazowanej palety

Rysunek 5 pozwala na wyznaczenie współrzędnej w osi Z dla Ŝądanej odległości szczęk uchwytu

od  powierzchni  zabazowanej  palety.  Na rysunku  5  poszczególne  oznaczenia  mają  następujące

znaczenia:

● X, Y, Z – układ współrzędnych,

● m,n – wartości współrzędnych odpowiednio w osiach x i y,

● q – zadana odległości q chwytaka od powierzchni zabazowanej palety,

● ki lub kl – znana wysokość chwytaka

● zq – poszukiwana współrzędna w osi z

zq=80�[207,5�qk i]

zq=80�207,5qk i

zq=qk1�127,5

Na rysunku 6 przedstawiono wymiary i wzajemne zaleŜności geometryczne palet na stanowiskach
montaŜowych. 

Rys. 6

W  celu  uchwycenia  przez  chwytak  najniŜszej  tulei  spoczywającej  bezpośrednio  w  gnieździe
zabazowanej palety naleŜy opuścić się z zamkniętymi szczękami wewnętrznymi nad odpowiednim gniazdem
na współrzędną  w osi  3 wynoszącą  2.5 (kropka).  Umieszczenie korka w tej tulei  wymaga opuszczenia
chwytaka z elementem w zamkniętych szczękach na wysokość 14 w osi 3 (nad tym gniazdem).

Przykład 4.9

# podprogram pobranie chwytaka prawego
:chap_prawy
write "Pobranie chwytaka prawego z pozycji gotow do pozycji gotow"
newl
# przejscie do nowej linii
write "Osiaganie pozycji - W gore"
robotgo 3 80
write ", w bok"
robotgo 1 100
write ", do tylu"
robotgo 2 0
write " - pozycja bezpieczna"
newl
write "W prawo po chwytak"
robotgo 1 650
write ", w dol"
robotgo 3 0
write " na dole."
newl
write "Pobierz chwytak..."
set 7
write "uchwycony."
newl
write "Do przodu"
robotgo 2 15
write ", wyjazd ze sprezyn zakonczony."
newl
write "W gore"
robotgo 3 80
write ", do przodu"
robotgo 2 80
write ", w lewo"
robotgo 1 100
write ", do tylu."
robotgo 2 0
write " Robot w pozycji bezpiecznej"
return

Współrzędne pozycji bezpiecznej robota wynoszą: 100, 0, 80. W tej pozycji naleŜy ustawiać robota w
czasie przerw w jego wykorzystywaniu. Pozycja  bezpieczna jest takŜe zalecana jako początek i  koniec
podprogramów z wykorzystaniem robota. Przed zakończeniem programu naleŜy odłoŜyć  pobrany chwytak
na wolny wieszak.
Uwaga: System symulatora ma inną przestrzeń roboczą.

4.3 Odczyt stanów czujników stanowiska

Instrukcja get odczytuje stan portu wejściowego wskazanego przez wyrazenie i zapisuje ja w zmiennej
zmienna. Składnia jej jest następująca:

get zmienna wyrazenie

gdzie:

zmienna przechowuje stan portu

wyrazenie wyznacza numer portu wejściowego (0-23)

Przykład 4.10

# definicja zmiennej ld03 (nadanie wartosci 2, rozna od odczytanej 1 lub 0)
let ld03 2
# odczyt portu 23 do zmiennej ld03
get ld03 23

Spis czujników w zaleŜności od numeru portu:

0 paleta na stacji załadowczo-wyładowczej

1 wieszak prawy zajęty,

2 wieszak środkowy zajęty,

3 wieszak lewy zajęty,

4 chwytak wewnętrzny,

5 czujnik prawy chwytaka daje sygnał – otwarty,

8 chwytak trzeci,

9 czujnik lewy chwytaka daje sygnał – zamknięty,

10 chwytak zewnętrzny

12 paleta przy zderzaku końcowym z lewej (stanowisko odkładcze),

13 przycisk start stacji załadowczo-wyładowczej,

14 paleta na stacji kodowania,

15 paleta na stacji odczytu,

16 trzpień kostki kodowej 1 z 4 lewy / górny (działa na przemian z 2),

17 trzpień kostki kodowej 2 z 4 lewy / dolny (działa na przemian z 1)

18 trzpień kostki kodowej 3 z 4 prawy / górny (działa na przemian z 4)

19 trzpień kostki kodowej 4 z 4 prawy / dolny (działa na przemian z 3)

20 wjazd na stacje Prawa

21 paleta na stacji Prawej w dole

22 wjazd na stacje Lewa

23 paleta na stacji Lewej w dole

Zmienna zmienna musi być wcześniej zdeklarowana przez wystąpieniem w instrukcji get . Patrz (3.1)
Dla  ułatwienia  i  ujednolicenia  zmiennych  odczyt  moŜe  następować  poprzez  podprogram

ld_czyt.prg przypisujący do zmiennych ldNN stany portów wejściowych “0” lub “1” gdzie NN to numer
zmiennych:

ld01 paleta na "lewy koniec",

ld02 wjazd na "stacja Lewa" zmiana stanu "1" na "0",

ld03 paleta na "stacja Lewa",

ld04 wjazd na "stacja Prawa" zmiana stanu "1" na "0",

ld05 paleta na "stacja Prawa",

ld06 paleta na "stacja odczytu",

ld07 kod 1 z 4 lewy / górny (działa na przemian z 2),

ld08 kod 2 z 4 lewy / dolny (działa na przemian z 1),

ld09 kod 3 z 4 prawy / górny (działa na przemian z 4),

ld10 kod 4 z 4 prawy / dolny (działa na przemian z 3),

ld11 paleta na "stacja kodowania",

ld12 paleta na "stacja załadowcza",

ld13 "wieszak prawy" zajęty,

ld14 "wieszak środkowy" zajęty,

ld15 "wieszak lewy" zajęty,

ld16 chwytak sygnał "zamknięty",

ld17 chwytak sygnał "otwarty",

ld18 przycisk "start" "stacja załadowcza " naciśnięty,

ld19 kod chwytaka wewnętrznego,

ld20 kod chwytaka bez nakładek szczęk,

ld21 kod chwytaka zewnętrznego

Podprogram  ld_czyt.prg przypisujący  stany  wszystkich  portów  wejściowych  do  zmiennych
wymaga aby przed jego pierwszym uŜyciem wywołać  podprogram  ld_def.prg definiujący zmienne. W
przypadku ruchu palet aby wychwycić  występowanie krótkich czasowo sygnałów z jedynie z  łączników
drogowych dotyczącego systemu transportowego zaleca się korzystać z podprogram ld_trans.prg . Dzięki
zmniejszeniu ilości sprawdzanych portów podprogram wykonuje się szybciej. W krańcowym przypadku, tj.
uŜycia wewnątrz pętli duŜej ilości instrukcji, zaleca się uŜycia pojedynczej komendy get wykonywanej duŜo
szybciej  w  stosunku  do  wywoływania  odpowiedniego  podprogramu  sczytującego  stan  wielu  portów
wejściowych.

4.4 Opis portów wyj ściowych

Instrukcje sterujące względem numer portu wyjściowego:

0 stacja robocza Lewa i Prawa uniesione (palety pozycjonowane),

1 wolne,

2 zderzak stacji kodującej w górze,

3 zderzak stacji prawej w górze,

4 zderzak stacji lewej w górze,

5 zderzak stacji odkładczej w górze (aktywny z lewej strony),

6 otwórz szczeki chwytaka zamocowanego w kiści robota,

7 chwytak zamocowany w kiści (przez zmieniacz chwytaków) robota,

8 kodowanie palety - trzpień 3 wysunięty,

9 kodowanie palety - trzpień 4 wysunięty,

10 kodowanie palety - trzpień 1 wysunięty,

11 kodowanie palety - trzpień 2 wysunięty,

12 zderzak stacji odczytu w dole (paleta nie zatrzymywana),

13 ruch palet w prawo,

14 ruch palet w lewo (z 13 uniesienie stacji załadowczo-wyładowczej)

5. Uwagi

Napisany  program  przed  uruchomieniem  na  stanowisku  montaŜowym  naleŜy  sprawdzić  w
symulatorze. Sprawdzenie odbywa się  ręcznie z uwagi na brak bezpośredniej moŜliwości jednoczesnego
uruchomienia kompilacji i symulatora.

Instrukcja  i  oprogramowania  symulatora  dostępnego  pod  adresem
http://www.cim.pw.edu.pl/monma_l/ odnoszą  się  do stanowiska montaŜowego przed modyfikacją  jego
usytuowania w budynku ST. W czasie zmiany lokalizacji we wrześniu 2003 nastąpiła reorganizacja, głównie
we współrzędnych robota. Z tego powodu w korzystaniu z tej instrukcji podczas pracy z symulatorem do
czasu  zmodernizowania  jego  oprogramowaniu  naleŜy  uwzględniać  poprawki  dostępne  w  instrukcji
symulatora. 

Wykorzystanie następujących procedur:

chwytak 
ruchpal
setkodpal
getkodpal
pauza

wymaga zastosowania biblioteki w swoim programie. Biblioteka zawarta jest w pliku robolib.prg, który
naleŜy dołączyć poleceniem:

include robolib.prg

ewentualnie z podaniem ścieŜki dostępu, np.

 include c:\programy\robolib.prg

Dyrektywa include powinna wystąpić na początku pliku.

6. FAQ Najcz ęściej zadawane pytania

Po zakończeniu programu paleta
przemieszcza się poza stację docelową

NaleŜy uwzględnić bezwładność układu transportowego
przy jego wyłączaniu.

Program działa ale kompilator zgłasza
błąd. W programie musi wystąpić instrukcja stop.

Podniesiony zderzak opuszcza się po
zakończeniu programu.

Koniec pracy kompilatora wymusza powrót stanowiska
EMM do stanu początkowego.

Instrukcja została przygotowana na podstawie: 
• "Język sterowania montaŜowego gniazda obróbkowego - gramatyka języka”  wersja 1.12 autorstwa

dr inŜ. Macieja HORCZYCZAKA ITM-PW
• “Biblioteka ROBOLIB” wersja 0.92 autorstwa dr inŜ. Macieja HORCZYCZAKA ITM-PW


